

Address Book Program

```

10 REM 'MAINPG'
20 REM *INITIL*
30 GOSUB 1000
40 REM *GREET*
50 GOSUB 3000
60 REM *CHOOSE*
70 GOSUB 3500
80 REM *EXECUT*
90 GOSUB 4000
100 IF CHOI <> 9 THEN 60
110 END
1000 REM *INITIL* SUBROUTINE
1010 GOSUB 1100: REM *CREARR* (CREATE ARRAYS) SUBROUTINE
1020 GOSUB 1400: REM *RDINFL* (READ IN FILE) SUBROUTINE
1030 GOSUB 1600: REM *SETFLG* (SET FLAGS) SUBROUTINE
1040 REM
1050 REM
1060 REM
1070 REM
1080 REM
1090 RETURN
1100 REM *CREARR* (CREATE ARRAYS) SUBROUTINE
1110 DIM NAMFLD$(50)
1120 DIM MODFLD$(50)
1130 DIM STRFLD$(50)
1140 DIM TWFLD$(50)
1150 DIM CNTFLD$(50)
1160 DIM TELFLD$(50)
1170 DIM NDXFLD$(50)
1180 REM
1190 REM
1200 REM
1210 LET SIZE = 0
1220 LET RMOD = 0
1230 LET SRD = 1
1240 LET CURR = 0
1250 REM
1260 REM
1270 REM
1280 REM
1290 REM
1300 RETURN
1400 REM *RDINFL* SUBROUTINE
1410 OPEN "I",#1,"ABK.DAT"
1420 INPUT #1,TEST$
1430 IF TEST$ = "@FIRST" THEN GOTO 1530: REM CLOSE AND RETURN
1440 LET NAMFLD$(1) = TEST$
1450 INPUT #1,MODFLD$(1),STRFLD$(1),TWFLD$(1),CNTFLD$(1),TELFLD$(1)
1460 INPUT #1,NDXFLD$(1)
1470 LET SIZE = 2
1480 FOR L = 2 TO 50
1490 INPUT #1,NAMFLD$(L),MODFLD$(L),STRFLD$(L),TWFLD$(L),CNTFLD$(L)
1500 INPUT #1,TELFLD$(L),NDXFLD$(L)
1510 LET SIZE = SIZE + 1
1520 IF EOF(1) = -1 THEN LET L = 50
1530 NEXT L
1540 CLOSE #1
1550 RETURN
1600 REM *SETFLG* SUBROUTINE
1610 REM SETS FLAGS AFTER *RDINFL*
1620 REM
1630 REM
1640 IF TEST$ = "@FIRST" THEN LET SIZE = 1
1650 REM
1660 REM
1670 REM
1680 REM
1690 RETURN
3000 REM *GREET* SUBROUTINE
3010 PRINT CHR$(12):REM CLEAR SCREEN
3020 PRINT
3030 PRINT
3040 PRINT
3050 PRINT
3060 PRINT TAB(12);"*WELCOME TO THE*"
3070 PRINT TAB(9);"*HOME COMPUTER COURSE*"
3080 PRINT TAB(6);"*COMPUTERISED ADDRESS BOOK*"
3090 PRINT
3100 PRINT TAB(5);"(PRESS SPACE-BAR TO CONTINUE)"
3110 FOR L = 1 TO 1
3120 IF INKEY$ <> " " THEN L = 0
3130 NEXT L
3140 PRINT CHR$(12)
3150 RETURN
3500 REM *CHOOSE* SUBROUTINE
3510 REM
3520 IF TEST$ = "@FIRST" THEN GOSUB 3860: REM *FIRST* SUBROUTINE
3530 IF TEST$ = "@FIRST" THEN RETURN
3540 REM 'CHMENU'
3550 PRINT CHR$(12)
3560 PRINT "SELECT ONE OF THE FOLLOWING"
3570 PRINT
3580 PRINT
3590 PRINT
3600 PRINT "1. FIND RECORD (FROM NAME)"
3610 PRINT "2. FIND NAMES (FROM INCOMPLETE NAME)"
3620 PRINT "3. FIND RECORDS (FROM TOWN)"
3630 PRINT "4. FIND RECORD (FROM INITIAL)"
3640 PRINT "5. LIST ALL RECORDS"
3650 PRINT "6. ADD NEW RECORD"
3660 PRINT "7. CHANGE RECORD"
3670 PRINT "8. DELETE RECORD"
3680 PRINT "9. EXIT & SAVE"
3690 PRINT
3700 PRINT
3710 REM 'INCHOI'
3720 REM
3730 LET L = 0
3740 LET I = 0
3750 FOR L = 1 TO 1
3760 PRINT "ENTER CHOICE (1 - 9)"
3770 FOR I = 1 TO 1

```

```

3780 LET A$ = INKEY$
3790 IF A$ = "" THEN I = 0
3800 NEXT I
3810 LET CHOI = VAL(A$)
3820 IF CHOI < 1 THEN L = 0
3830 IF CHOI > 9 THEN L = 0
3840 NEXT L
3850 RETURN
3860 REM *FIRST* SUBROUTINE (DISPLAY MESSAGE)
3870 LET CHOI = 6
3880 PRINT CHR$(12): REM CLEAR SCREEN
3890 PRINT
3900 PRINT TAB(8);"THERE ARE NO RECORDS IN"
3910 PRINT TAB(8);"THE FILE. YOU WILL HAVE"
3920 PRINT TAB(6);"TO START BY-ADDING A RECORD"
3930 PRINT
3940 PRINT TAB(5);"(PRESS SPACE-BAR TO CONTINUE)"
3950 FOR B = 1 TO 1
3960 IF INKEY$ <> " " THEN B = 0
3970 NEXT B
3980 PRINT CHR$(12): REM CLEAR SCREEN
3990 RETURN
4000 REM *EXECUT* SUBROUTINE
4010 REM
4020 REM
4030 REM
4040 IF CHOI = 1 THEN GOSUB 13000: REM *FNDREC*
4050 REM 2 IS *FNDNMS*
4060 REM 3 IS *FNDTWN*
4070 REM 4 IS *FNDINT*
4080 REM 5 IS *LSTREC*
4090 IF CHOI = 6 THEN GOSUB 10000: REM *ADREC*
4100 IF CHOI = 7 THEN GOSUB 14000: REM *MODREC*
4110 IF CHOI = 8 THEN GOSUB 15000: REM *DELREC*
4120 IF CHOI = 9 THEN GOSUB 11000: REM *EXPROC*
4130 REM
4140 RETURN
10000 REM *ADREC* SUBROUTINE
10010 PRINT CHR$(12): REM CLEAR SCREEN
10020 INPUT "ENTER NAME";NAMFLD$(SIZE)
10030 INPUT "ENTER STREET";STRFLD$(SIZE)
10040 INPUT "ENTER TOWN";TWFLD$(SIZE)
10050 INPUT "ENTER COUNTY";CNTFLD$(SIZE)
10060 INPUT "ENTER TELEPHONE NUMBER";TELFLD$(SIZE)
10070 LET RMOD = 1: LET SRD = 0: REM MODIFIED & NOT SORTED
10080 LET NDXFLD$(SIZE) = STR$(SIZE)
10090 LET TEST$ = ""
10100 COSUB 10200: REM *MODNAM*
10110 LET CHOI = 0
10120 LET SIZE = SIZE + 1
10130 REM
10140 REM
10150 RETURN
10200 REM *MODNAM* ROUTINE
10210 REM CONVERTS CONTENTS OF NAMFLD$ TO UPPER CASE,
10220 REM REMOVES RUBBISH, AND STORES IN THE ORDER:
10230 REM SURNAME+SPACE+FORENAME IN MODFLD$
10240 REM
10250 LET N$ = NAMFLD$(SIZE)
10260 FOR L = 1 TO LEN(N$)
10270 LET TEMP$ = MID$(N$,L,1)
10280 LET T = ASC(TEMP$)
10290 IF T >= 97 THEN T = T - 32
10300 LET TEMP$ = CHR$(T)
10310 LET P$ = P$ + TEMP$
10320 NEXT L
10330 LET N$ = P$
10340 REM LOCATE LAST SPACE
10350 FOR L = 1 TO LEN(N$)
10360 IF MID$(N$,L,1) = " " THEN S = L
10370 NEXT L
10380 REM REMOVE RUBBISH AND STORE FORENAME
10390 REM IN CNAM$
10400 FOR L = 1 TO S - 1
10410 IF ASC(MID$(N$,L,1)) > 64 THEN CNAM$ = CNAM$ + MID$(N$,L,1)
10420 NEXT L
10430 REM REMOVE RUBBISH AND STORE SURNAME
10440 REM IN SNAM$
10450 FOR L = S + 1 TO LEN(N$)
10460 IF ASC(MID$(N$,L,1)) > 64 THEN SNAM$ = SNAM$ + MID$(N$,L,1)
10470 NEXT L
10480 LET MODFLD$(SIZE) = SNAM$ + " " + CNAM$
10490 LET P$ = " ": LET N$ = " ": LET SNAM$ = " ": LET CNAM$ = " ":
LET S = 0
10500 RETURN
11000 REM *EXPROC* SUBROUTINE
11010 REM SORTS AND SAVES FILE
11020 REM IF ANY RECORD HAS BEEN
11030 REM MODIFIED (RMOD = 1)
11040 REM OR NOT SORTED (SRD = 0)
11050 REM RMOD = 0 AND SRD = 0 IS ILLEGAL
11060 REM
11070 IF RMOD = 0 AND SRD = 1 THEN RETURN
11080 IF RMOD = 1 AND SRD = 0 THEN GOSUB 11200: REM *SRTREC*
11090 GOSUB 12000: REM *SAVREC*
11100 RETURN
11200 REM *SRTREC* SUBROUTINE
11210 REM SORTS ALL RECORDS BY MODFLD$ INTO
11220 REM ALPHABETICAL ORDER AND UPDATES NDXFLD
11230 REM
11240 REM
11250 LET S = 0
11260 FOR L = 1 TO SIZE - 2
11270 IF MODFLD$(L) > MODFLD$(L + 1) THEN GOSUB 11350
11280 NEXT L
11290 IF S = 1 THEN 11250
11300 REM
11310 REM
11320 LET SRD = 1: REM SETS 'FILE SORTED' FLAG
11330 REM
11340 RETURN
11350 REM *SWPREC* SUBROUTINE
11360 LET TNAMFLD$ = NAMFLD$(L)
11370 LET TMODFLD$ = MODFLD$(L)
11380 LET TSTRFLD$ = STRFLD$(L)

```